** Clearing Stata memory
capture log close
clear all
set more off, perm
set seed 1234

///////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////
//////////////////////////// Table O.21: Priority Subjects, Most Difficult Questions and Performance ///////////////////////
///////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////

** Opening Phase 2 norm_scores dataset 
use "Work Data/Gender_Phase2_long.dta",clear

*** Creating variables
encode subject, gen (sub)
tab subject, gen (d_sub)
label var sub "Subject"

** Subject dummies
rename d_sub1 Biology
rename d_sub2 Chemistry
rename d_sub3 Geography
rename d_sub4 History
rename d_sub5 Language
rename d_sub6 Mathematics
rename d_sub7 Physics
rename d_sub8 Portuguese
* Labels
label var Biology "Biology"
label var Chemistry "Chemistry"
label var Geography "Geography"
label var History "History"
label var Math "Mathematics"
label var Physics "Physics"
label var Portuguese "Portuguese"
label var Language "Foreign Language"

** Interaction: priority X female
gen fem_priority=female*priority
label var fem_priority "Female $\times$ Priority"

** Interaction: priority X subject
foreach v of varlist Biology-Portuguese {
gen fem_`v'=`v'*female
label var fem_`v' "Female $\times$ `v'"
gen prio_`v'=priority*`v'
label var prio_`v' "Priority $\times$ `v'"
gen fem_prio_`v'=fem_priority*`v'
label var fem_prio_`v' "Female $\times$ Priority $\times$ `v'"
}

global subject "Chemistry Geography History Mathematics Physics"
global subject_fem "fem_Chemistry fem_Geography fem_History fem_Mathematics fem_Physics"

** P1 scores: P1 normalized subject-specific scores
forvalues i=2(1)4 {
gen norm_p1score`i'=norm_p1score^`i'
sum norm_p1score`i'
}

*********************************************************************************
****************   Relative performances ****************************************
*********************************************************************************

** ENEM

foreach v in  norm_enem_w {
bys year female: egen `v'_ave_g=mean(`v')
gen `v'_g=`v'-`v'_ave_g
bys year female: sum `v'_g
}
drop  norm_enem_w_ave_g

* Interaction: subject X ENEM
foreach v of varlist Biology-Portuguese {
gen enem_`v'=`v'*norm_enem_w_g
label var enem_`v' "ENEM $\times$ `v'"
gen fem_enem_`v'=female*norm_enem_w_g*`v'
label var fem_enem_`v' "Female $\times$ ENEM $\times$ `v'"
forvalues i=2(1)4 {
gen enem_`v'_`i'=enem_`v'^`i'
gen fem_enem_`v'_`i'=fem_enem_`v'^`i'
}
sum enem_`v'* fem_enem_`v'*
}

global g_pol_enem_sub "enem_Chemistry* enem_Geography* enem_History* enem_Mathematics* enem_Physics*"
d $g_pol_enem_sub

* Priority x relative performance in ENEM:
foreach v in norm_enem_w {
gen `v'_priority_g=`v'_g*priority
forvalues i=2(1)4 {
gen `v'_priority_g`i'=`v'_g^`i'*priority
sum `v'_priority_g`i'
}
}

global g_norm_enem_w_prio norm_enem_w_priority_g*
d $g_norm_enem_w_prio

** Phase 1 scores

foreach v in norm_p1score {

tab year, sum(`v')
bys year subject female: egen gs_`v'_ave=mean(`v')
gen gs_`v'=`v'-gs_`v'_ave
bys year female subject: sum gs_`v'
drop gs_`v'_ave

forvalues i=2(1)4 {
gen gs_`v'`i'=gs_`v'^`i'
sum gs_`v'`i'
}

global gs_pol_`v' gs_`v' gs_`v'2 gs_`v'3 gs_`v'4
d $gs_pol_`v'

* Priority x Phase 1 scores:
gen gs_`v'_prio=gs_`v'*priority
forvalues i=2(1)4 {
gen gs_`v'_prio`i'=gs_`v'`i'*priority
sum gs_`v'_prio*
}
}

global gs_pol_norm_p1score_prio gs_norm_p1score_prio*
d $gs_pol_norm_p1score_prio

*********************************************************************************
**************** Main sample ****************************************************
*********************************************************************************

* 1) Only years before the affirmative action took place
drop if aa_year==1
drop if year==2000
tab year

* 2) Drop Portuguese and Foreign Language (in Phase 1 there is no Portuguese or Foreign Language exams - For Portuguese Phase 1 has an essay)
 tab subject, sum(norm_p1score)
 drop if subject=="lang" | subject=="port" 
 tab subject, sum(norm_p1score)
 drop Language Portuguese prio_Language prio_Portuguese fem_prio_Language fem_prio_Portuguese 

*********************************************************************************
**************** Scores by order ************************************************
*********************************************************************************

forvalues i=1(1)12 {
gen score_item`i'=.
levelsof subject, local(levels) 
foreach s of local levels {
replace score_item`i'=`s'`i'_st2 if subject=="`s'"
}
tab subject, sum(score_item`i')
}

***** Keeping only variables that will be used in the regression
keep  subject year score_item* perfect_score_item* skipped_item* zero_score* female priority fem_priority   $g_pol_enem_sub ///
p1score norm_enem_w  inscri2 norm_score totalzeros zeros1to4 zeros9to12  maxtotal max1to4 max9to12 totalmissing missing1to4  missing9to12  ///
$subject $subject_fem $gs_pol_norm_p1score $gs_pol_norm_p1score_prio $g_norm_enem_w_prio ratio_first_last4 diff_first_last4 norm_enem_w_g

*********************************************************************************
**************** RESHAPE  *******************************************************
*********************************************************************************

* Reshaping data - Student x subject x item

egen id_sub=group(inscri2 subject)

reshape long score_item perfect_score_item skipped_item zero_score, i(id_sub) j(order)

*********************************************************************************
**************** Measures of item difficulty   **********************************
*********************************************************************************

**** Measures of item easiness/difficulty - Based on Iriberri & Rey-Biel (2019), Economic Journal

* Average score in a question/item
bys year subject order: egen avg_score=mean(score_item)
tab subject, sum(avg_score)
tab year, sum(avg_score)
tab order,sum(avg_score)
label var avg_score "Question's average score"

* Difficulty dummies

bys year subject: egen min_avg_score=min(avg_score)
gen most_difficult=1 if avg_score==min_avg_score
replace  most_difficult=0 if avg_score>min_avg_score
tab most_difficult
tab most_difficult, sum(avg_score)
tab most_difficult, sum(score_item)

*********************************************************************************
**************** Regressions   **************************************************
*********************************************************************************

*** Most difficult

gen fem_mdifficult=female*most_difficult
label var fem_mdifficult "Female $\times$ Most difficult question"
global most_difficult fem_mdifficult most_difficult

 **************************** Difficulty level and priority subjects ****************************
 
gen prio_mdifficult=priority*most_difficult
label var prio_mdifficult "Priority $\times$ Most difficult question"
gen fem_prio_mdifficult=female*most_difficult*priority
label var fem_prio_mdifficult "Female $\times$ Priority $\times$ Most difficult question"
global most_difficult_priority "fem_prio_mdifficult prio_mdifficult most_difficult fem_mdifficult"

foreach x in most_difficult_priority {

estimates clear

reghdfe score_item female priority fem_priority $`x' norm_enem_w_g, cluster(inscri2) absorb(order)
estimates store reg1
estadd ysumm
reghdfe score_item female priority fem_priority $`x' norm_enem_w_g  $subject $subject_fem, cluster(inscri2)  absorb(order)
estimates store reg2
reghdfe score_item priority fem_priority $`x' $subject $subject_fem, cluster(inscri2) absorb(order inscri2)
estimates store reg3
reghdfe score_item priority fem_priority $`x' $subject $subject_fem $g_pol_enem_sub, cluster(inscri2) absorb(order inscri2)
estimates store reg4
reghdfe score_item priority fem_priority $`x' $subject $subject_fem $g_pol_enem_sub $g_norm_enem_w_prio, cluster(inscri2) absorb(order inscri2)
estimates store reg5
reghdfe score_item priority fem_priority $`x' $subject $subject_fem $g_pol_enem_sub $g_norm_enem_w_prio $gs_pol_norm_p1score, cluster(inscri2) absorb(order inscri2)
estimates store reg6
reghdfe score_item priority fem_priority $`x' $subject $subject_fem $g_pol_enem_sub $g_norm_enem_w_prio $gs_pol_norm_p1score $gs_pol_norm_p1score_prio, cluster(inscri2) absorb(order inscri2)
estimates store reg7

estadd local order "Yes":  reg*

estadd local sub_fe "No":  reg1
estadd local sub_fe "Yes": reg2 reg3 reg4 reg5 reg6 reg7

estadd local subgender_fe "No":  reg1
estadd local subgender_fe "Yes": reg2 reg3 reg4 reg5 reg6 reg7

estadd local ind_fe "No": reg1 reg2 
estadd local ind_fe "Yes": reg3 reg4 reg5 reg6 reg7

estadd local enemsub "No":  reg1 reg2 reg3 
estadd local enemsub "Yes": reg4 reg5 reg6 reg7

estadd local enemprio_pol4 "No":  reg1 reg2 reg3 reg4 
estadd local enemprio_pol4 "Yes": reg5 reg6 reg7

estadd local p1score_pol4 "No": reg1 reg2 reg3 reg4 reg5
estadd local p1score_pol4 "Yes": reg6 reg7

estadd local p1scoreprio_pol4 "No": reg1 reg2 reg3 reg4 reg5 reg6
estadd local p1scoreprio_pol4 "Yes": reg7 

* Tex
esttab reg1 reg2 reg3 reg4 reg5 reg6 reg7 using "Output/p_Female_`x'_score.tex", se star(* 0.10 ** 0.05 *** 0.01) nogap ///
stats(ymean ysd sep N N_clust newline order sub_fe subgender_fe ind_fe enemsub enemprio_pol4 p1score_pol4 p1scoreprio_pol4, /// 
fmt(%3.2fc %3.2fc %1s %9.0fc %9.0fc %1s %3s %3s %3s %3s %3s %3s %3s) labels("Mean dependent variable" " Std.dev dependent variable" " " /// 
"Number of observations"  "Number of applicants" " "  "Order FE"  "Subject FE" "Subject-gender FE"  "Individual FE" "ENEM $\times$ Subject FE" "ENEM $\times$ Priority" "Phase 1 scores" "Phase 1 scores $\times$ Priority")) b(%7.3f) se(%7.3f)  replace f label nomtitle collabels(none) booktabs ///
 refcat(female " \\ \multicolumn{8}{l}{\textit{Dependent variable: Questions' raw scores (Phase 2)}} \\", nolabel)  keep(female priority fem_priority $`x' norm_enem_w_g)
 
}
 